

LABORATOR 4

Analiza de curent alternativ folosind Spice. Filtre analogice

1. Ce este analiza de curent alternativ?

Programul Spice poate fi folosit pentru analiza de curent alternativ - AC (fie că este vorba de o singură frecvență, fie că este vorba de un domeniu de frecvențe). Pentru obținerea rezultatelor în cadrul analizei de curent continuu este nevoie de tandemul comenzilor **.DC** și **.PRINT DC**. Acum este o situație similară folosind comenzile **.AC** și **.PRINT AC**.

Programul Spice calculează în modul de simulare de curent alternativ *răspunsul în frecvență al circuitelor liniare și al celor echivalente de semnal mic*. Programul Spice folosește pentru analiza de curent alternativ metoda nodală modificată în complex și operează cu fazori.

Până acum toate sursele de tensiune și de curent folosite erau în curent continuu. Sintaxa surselor AC este asemănătoare.

Informațiile ce trebuiesc furnizate sunt:

- numele sursei care trebuie să înceapă cu V sau I;
- nodurile între care este conectată;
- amplitudinea (valoarea maximă) (X_{\max});
- faza sa inițială (α).

Separat, frecvența *tuturor* surselor din circuit este definită printr-o comandă **".AC"**.

Exemple de definiții de surse de curent alternativ:

```
*nume lista nodurilor tip valoare fază(grade)
Vlac 1 0 AC 230V 45
Vba 2 3 AC 240; fază inițială 0 grade
I3y 3 4 AC 10.0A -45; fază inițială -45 grade
Is1 1 9 AC 35mA; 35 miliamperi la 0 grade
```

Specificarea tipului AC este obligatorie la definirea surselor deoarece cel implicit este DC. Faza este măsurată în grade. Dacă nu se precizează faza, aceasta se presupune 0. În exemplele anterioare V și A (ca unitați de măsură) sunt opționale după valorile tensiunii și ale curentului.

Rezistorul se declară la fel ca în curent continuu:

```
R_num N1 N2 valoare în Ohm
```

Condensatorul liniar se declară în Spice astfel:

```
C_num N1 N2 valoare în Farad
```

Bobina liniară se declară în Spice astfel:

```
L_nume N1 N2 valoare în Henri
```

Pentru a putea folosi comanda **.PRINT AC**, trebuie activată comanda **.AC**.

Comanda **.AC** este folosită pentru a parcurge un domeniu de frecvențe pentru un circuit dat. Aceasta corespunde găsirii răspunsului în frecvență. Există 3 tipuri de parcurgere a domeniului frecvență: **LIN**, **DEC** și **OCT**.

Exemple de comenzi **.AC**:

```
.AC LIN 1 50Hz 50Hz  
.AC LIN 11 150 250; o baleiere liniară a domeniului  
.AC DEC 20 10Hz 100kHz; o baleiere pe decade a domeniului
```

Prima comandă realizează o analiză numai pentru frecvența de 50Hz (am ales domeniul liniar). Adăugarea unităților Hz după valoare este opțională. Cea de-a doua comandă parcurge domeniul format din frecvențele 150Hz, 160Hz, 170Hz, 180Hz, 190Hz, 200Hz, 210Hz, 220Hz, 230Hz, 240Hz și 250Hz. Cea de-a treia comandă parcurge logaritmically domeniul folosind câte 20 de puncte pe decadă pe un domeniu de 4 decade (10Hz-100Hz, 100Hz-1kHz, 1kHz-10kHz și 10kHz-100kHz).

Afișarea componentelor fazoriale (numere complexe) necesită anumite opțiuni pentru comanda **.PRINT AC**. Există 4 expresii necesare pentru aceasta: modul, fază („phase” în engleză), parte reală și parte imaginară. Pot fi afișate tensiuni și curenți. De exemplu, pentru a afișa modulul unei tensiuni dintre nodurile 2 și 3, vom specifica **VM(2,3)**, sau simplu **V(2,3)**. Faza în grade pentru aceeași tensiune va fi **VP(2,3)** – P provine de la „Phase”. Dacă dorim modulul curentului prin rezistorul $R_{sarcina}$, se specifică **IM(Rsarcina)**. Partea reală a tensiunii la nodul 7 va fi **VR(7)**, iar partea sa imaginară va fi **VI(7)**. Ca și în cazul comenzii **.PRINT DC**, nu există un număr limită de câte ori poate fi folosită comanda.

Exemple:

```
.PRINT AC VM(30,9) VP(30,9); modulul și faza tensiunii  
.PRINT AC IR(Rs) II(Rs); părțile reală și imaginară ale  
curentului prin Rs  
.PRINT AC VM(13) VP(13) VR(13) VI(13); toate informațiile la  
nodul 13
```

Pentru afișarea unor grafice în funcție de frecvență, se folosește comanda **.PROBE**. Pentru a vizualiza diferite variații, folosim din meniu *Trace->Add Traces* de unde putem alege fie mărimea respectivă, fie diferite funcții ale mărimii. De exemplu pentru variația fazelor se alege funcția P(), pentru variația părții imaginare se alege funcția IMG(), iar pentru variația părții reale se alege funcția R().

2. Circuit R-L serie

În figura 3.1 este prezentat un circuit format dintr-o rezistență $R_I=10\Omega$ și o bobină $L_I=10mH$, legate în serie. Combinația poate reprezenta, de exemplu, cazul unei bobine reale.

Se calculează curentul prin circuit pentru un semnal sinusoidal aplicat la intrare V_{in} , cu valoarea maximă $1V$ și frecvență $f=100Hz$.




Fig. 3.1.Circuit R-L serie

Se folosește următorul program:

```
LTspice XVII - [circuit R-L serie]
File Edit View Simulate Tools Window
Circuit R-L serie
Vin 1 0 AC 1
R1 1 2 10
L1 2 0 10m
.AC LIN 1 100 100
.PRINT AC I(R1)
.END
```

În urma rulării programului se obține următorul rezultat:

```
Circuit R-L serie
--- AC Analysis ---
frequency: 100 Hz
I(R1): mag: 0.0846672 phase: -32.1393° device_current
```

Observații:

Frecvența de analiză este de 100Hz, amplitudinea curentului prin circuit este de 84mA și este defazat în urma tensiunii cu 32° .

Aceste rezultate permit calcularea mărimilor de mai jos.

Impedanța circuitului:

$$Z = V_{in}/I(R_1) = 11.9 * \exp(j * 32^\circ)$$

Puterile activă, reactivă și aparentă:

$$\begin{aligned} P &= U * I * \cos\varphi = 1V * 0.084A * \cos 32^\circ = 0.0712W \\ Q &= U * I * \sin\varphi = 1V * 0.084A * \sin 32^\circ = 0.0445VAR \\ S &= U * I = 1V * 0.084A = 0.084VA \end{aligned}$$

3. Circuit R-C serie

Circuitul R-C serie cu valorile $R_1=10\Omega$ și $C_1=100\mu F$ este prezentat în figura 3.2. Circuitul este analizat la frecvență $f=318\text{Hz}$.




Fig. 3.2.Circuit R-C serie

Se folosește următorul program:

```
Circuit R-C serie
Vin 1 0 AC 1
R1 2 0 10
C1 1 2 100u
.AC LIN 1 318 318
.PRINT AC I(R1) V(2,0) V(1,2)
.END
```

În urma rulării programului se obține următorul rezultat:

--- AC Analysis ---					
frequency:	318	Hz			
V(1):	mag:	1	phase:	-3.18055e-015°	voltage
V(2):	mag:	0.894253	phase:	26.5874°	voltage
I(R1):	mag:	0.0894253	phase:	26.5874°	device_current

Folosind fișierul de ieșire se poate arăta că $V_{in} = \sqrt{V^2(R_1) + V^2(C_1)}$

4. Circuit R-L-C serie

Circuitul R-L-C serie este prezentat în figura 3. Comportarea sa în regim sinusoidal depinde de frecvență și putem distinge trei situații.




Fig. 3.3. Circuit R-L-C serie

a) *Rezonanța tensiunilor (comportare rezistivă)*

Aceasta are loc atunci când $U_L = U_C$ deci $X_L = X_C$. Aceasta se întâmplă la frecvența:

$$\omega = \omega_0 = \frac{1}{\sqrt{LC}}$$

La această frecvență circuitul are un caracter pur rezistiv, impedanța sa fiind minimă:

$$|Z| = |Z|_{\min} = \sqrt{R^2 + (X_L - X_C)^2} = R$$

iar curentul prin circuit are valoarea maximă egală cu:

$$I_{\max} = U/Z_{\min} = U/R$$

b) *Comportare capacitivă*

Pentru $\omega = \omega_1 < \omega_0$, avem $X_C > X_L$ și deci $U_C > U_L$ iar circuitul se va comporta ca un grup R-C echivalent și va determina un defazaj al curentului înaintea tensiunii la borne.

c) *Comportare inductivă*

Pentru $\omega = \omega_2 > \omega_0$, avem $X_L > X_C$ și deci $U_L > U_C$ iar circuitul se va comporta ca un grup serie R-L echivalent și va determina un defazaj al curentului în urma tensiunii la borne.

Considerând valorile $R = 100\Omega$, $L = 25.33mH$ și $C = 1\mu F$, frecvența de rezonanță are aproximativ valoarea $f_0 = 1000Hz$.

Se rulează următorul fișier LTSpice:

```

****
Circuit R-L-C serie
Iin 0 1 AC 1
R 1 2 100
L 2 3 25.33m
C 3 0 1u
RC 3 0 1meg
.AC LIN 1 1000 1000
.PRINT AC V(1)
.PRINT AC V(1,2) V(2,3) V(3)
.END

```

Linia de program **RC 3 0 1meg** a fost introdusă deoarece sursa de curent are rezistență internă infinită și astfel nodurile 1, 2 și 3 nu ar avea o cale spre masă (pentru a

verifica, se poate încerca varianta fără R_C). Rezistența $R_C = 1M\Omega$ introdusă în paralel cu condensatorul asigură această cale de curent fără a modifica practic funcționarea circuitului.

Circuit R-L-C serie					
--- AC Analysis ---					
frequency:	1000	Hz			
V(1):	mag:	100.026	phase:	-0.00106269°	voltage
V(2):	mag:	0.0263956	phase:	-4.03039°	voltage
V(3):	mag:	159.155	phase:	-89.9909°	voltage

Se rulează analiza de mai sus și pentru $f_1 = 500Hz$ și $f_2 = 2000Hz$ în vederea comparării rezultatelor.

Tot ca exercițiu, se poate schimba alimentarea într-o sursă de tensiune sinusoidală de amplitudine $1V$. Se observă că în acest caz rezistența R_C în paralel cu C nu mai este necesară. De ce?

5. Transferul maxim de putere

Transferul maxim de putere în curent continuu are loc pentru cazul în care rezistența de sarcină este egală cu rezistența internă a sursei echivalente de tensiune. În cazul circuitelor de curent alternativ, atât impedanța internă a sursei echivalente cât și impedanța de sarcină sunt mărimi complexe. Se poate demonstra că într-o astfel de situație, transferul maxim de putere are loc dacă impedanța de sarcină și impedanța internă a sursei sunt complex conjugate.




Fig. 3.4. Circuit studiat pentru transferul maxim de putere

Pentru circuitul din figura 3.4, de exemplu, V_{in} are amplitudinea $10V$ și frecvența $1kHz$, iar $R_i = 10\Omega$ și $L_i = 25.33mH$. Atunci pentru ca transferul să fie maxim, trebuie ca $R_L = 10\Omega$ și $C_L = 1\mu F$. De ce?

Se rulează următorul exemplu:

```
Transfer maxim de putere
Vin 1 0 AC 10
RI 1 2 10
LI 2 3 25.33M
RL 3 4 10
CL 4 0 1U
.AC LIN 1 1K 1K
.PRINT AC I(RL)
.END
```

În urma rulării programului se obține următorul rezultat:

```
Transfer maxim de putere
--- AC Analysis ---
frequency:      1000      Hz
I(RL):          mag:    0.499975 phase: 0.00532612°      device_current
```

Se observă că $I(R_L)=0.5A$ ceea ce conduce la o putere absorbită egală cu $2.5W$.

6. Exemple pentru domenii de frecvențe

6.1. Rezonanța RLC serie

Se analizează circuitul RLC serie din figura 3.5. În acest caz rezonanța tensiunilor are loc la frecvența: $f_0 = \frac{1}{2\pi\sqrt{LC}} = 999.3Hz$




Fig. 3.5. Circuit R-L-C serie

Se va analiza comportarea circuitului în gama de frecvențe cuprinse între 100Hz și 10kHz.

Se parcurg pașii de mai jos:


- Se rulează următorul program:

```


Rezonanta serie
Vin 1 0 AC 1
R 1 2 100
L 2 3 25.33m
C 3 0 1u
.AC LIN 1000 100Hz 10kHz
.PROBE
.END

```


- Se deschide automat editorul grafic:




- Se apasă click-dreapta și se selectează ADD Traces:



- Se selectează I(R):




- Rezultă modulul (cu linie continuă) și faza (cu linie întreruptă) ale curentului $I(R)$. De asemenea se observă maximum lui $I(R)$ la 1000Hz:




- Se trasează impedanța $Z(\Omega) = V(1)/I(R)$. Pentru început se dă click dreapta și se selectează Add Plot Pane:



- Apoi se adaugă graficul $Z(\Omega) = V(1)/I(R)$ (se observă modulul și faza):



- Pentru a activa cursorul, se apasă pe $V(1)/I(R)$ în partea de sus și se obține:



Activând cursorul ca în ultima imagine de mai sus, să se stabilească valoarea impedanței ($Z(\omega) = \frac{V(1)}{I(R)}$) și a fazei pentru 3 frecvențe: $f_1=500\text{Hz}$, $f_0=1000\text{Hz}$ și $f_2=2000\text{Hz}$.

Se remarcă că pentru $f < f_0$, circuitul se comportă capacitiv, iar pentru $f > f_0$, el se comportă inductiv.

6.2. Filtre pasive

Filtrele pasive, conținând numai rezistoare, bobine și condensatoare, își găsesc numeroase aplicații în practică. Un filtru analogic este de obicei un cuadripol (sau diport) cu o poartă de intrare și o poartă de ieșire. Caracteristicile filtrului (de modul și de fază) conduc la transformarea semnalului de intrare în semnal de ieșire. Distingem filtre trece jos (FTJ) care lasă să treacă doar componentele de frecvență mică, filtre trece sus (FTS) care lasă să treacă doar componentele de frecvență înaltă, filtre trece bandă (FTB), filtre oprește bandă (FOB), filtre pieptene (lasă să treacă doar anumite frecvențe).

Cele mai simple filtre se obțin prin înserierea unui rezistor cu un condensator.


6.2.1. Filtrul trece-jos și filtrul trece-sus

În figura 3.2 se află schema unui filtru trece jos dacă se consideră drept ieșire tensiunea pe condensator și un filtru trece sus dacă se consideră ieșirea tensiunea pe rezistor.


- Se folosește următorul program Spice:

```
Filtru trece-jos
Vin 1 0 AC 1
R 2 0 10k
C 1 2 10nF
.ac dec 10 1hz 1meghz
.probe
.end
```

- Se trasează ieșirea pe condensator V(1,2) (modul și fază):



- Se trasează și dreapta de valoare = 0.707 și se notează frecvența de intersecție cu graficul. Aceasta se numește frecvență de tăiere.



Se poate observa că la frecvența de tăiere faza este de -45° .

Ca un exercițiu, se propune și analizarea într-un mod asemănător a comportării filtrului trece-sus (ieșirea pe rezistor).

6.2.2. Filtrul trece-bandă de tip Butterworth

Schema unui filtru trece-bandă de tip Butterworth este prezentată în figura 3.6.





Fig. 3.6. Filtru trece-bandă de tip Butterworth

- Se folosește următorul program:

```
FILTRU_Butterworth
Filtru trece-banda
*de tip Butterworth
Vin 1 0 ac 1
l12 1 2 100u
c23 2 3 2u
c30 3 0 50u
l30 3 0 4u
r30 3 0 1
.ac dec 1000 5k 25k
.probe
.end
```

- Se afișează curba V(3). Banda de trecere se determină citind coordonatele punctelor de intersecție cu dreapta 0.707. Se obține $B=2.252kHz$.



6.2.3. Filtrul oprește-bandă

Schema unui filtru oprește-bandă este prezentată în figura 3.7.





Fig. 3.7. Filtru oprește-bandă

- Se folosește următorul program:

```
filtru-opreste-banda.cir
Filtru opreste-banda
Vin 1 0 AC 1
R12 1 2 10
R23 2 3 0.5
C34 3 4 26.5U
L40 4 0 265M
R20 2 0 10
.ac OCT 100 40 80
.probe
.end
```

- După rulare se afișează curba $V(2)$. Frecvențele de tăiere se determină citind coordonatele punctelor de intersecție cu dreapta $0.707 \cdot 500\text{mV}$ deoarece amplitudinea în benzile de trecere este de 500mV . Se obțin $f_1 = 58.4\text{Hz}$ și $f_2 = 61.7\text{Hz}$.




7. Probleme

Simulați problemele de mai jos folosind LTSpice.


7.1. Problema 1

Pentru circuitul de mai jos se consideră $f=50\text{Hz}$. Găsiți curenții (modul și fază) prin laturile circuitului.




7.2. Problema 2

Pentru circuitul de mai jos se consideră $f=50\text{Hz}$. Găsiți curenții (modul și fază) prin laturile circuitului.




7.3. Problema 3

Pentru circuitul de mai jos se consideră $f=50\text{Hz}$. Găsiți curenții (modul și fază) prin laturile circuitului.




7.4. Problema 4

Pentru circuitul de mai jos găsiți curenții (modul și fază) prin laturile circuitului și tensiunea la bornele sursei de curent:




7.5. Problema 5

Pentru circuitul de mai jos găsiți curenții (modul și fază) prin laturile circuitului și tensiunea la bornele sursei de curent:




7.6. Problema 6

Pentru circuitul de mai jos găsiți curenții (modul și fază) prin laturile circuitului și tensiunea la bornele sursei de curent:




7.7. Problema 7

Pentru circuitul de mai jos găsiți curenții (modul și fază) prin laturile circuitului și tensiunea la bornele sursei de curent:



7.8. Problema 8

Pentru circuitul de mai jos găsiți curenții (modul și fază) prin laturile circuitului și tensiunea la bornele sursei de curent:



7.9. Problema 9

Pentru circuitul de mai jos găsiți curenții (modul și fază) prin laturile circuitului și tensiunea la bornele sursei de curent:

